```
>
```

Repeated Real Eigenvalues

Repeated Eigenvalue with Linearly Independent Eigenvectors

```
> restart:with(LinearAlgebra):with(plots):with(DEtools):
```

We wish to solve the planar homogeneous linear system

$$\begin{bmatrix} \frac{dx}{dt} = 4\,x \\ \frac{dy}{dt} = 4\,y \end{bmatrix} \text{ or } \mathbf{r}' = A\mathbf{r} \text{ with } A = \begin{bmatrix} 4 & 0 \\ 0 & 4 \end{bmatrix}.$$

This is an **uncoupled** system. We begin with the matrix A.

```
> A:=Matrix(2,2,[[4,0],[0,4]]);
```

$$A := \begin{bmatrix} 4 & 0 \\ 0 & 4 \end{bmatrix}$$

We find its eigenvalues and corresponding eigenvectors.

```
> v:=Eigenvectors(A,output=list);
```

$$v := \left[\left[4,\ 2,\ \left\{\begin{bmatrix} 1 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 1 \end{bmatrix}\right\}\right]\right]$$

We see that 4 is the only eigenvalue and is of multiplicity two with $\begin{bmatrix} 0 \\ 1 \end{bmatrix}$ and $\begin{bmatrix} 1 \\ 0 \end{bmatrix}$ as corresponding eigenvectors. Thus the general solution is $\mathbf{r} = c_1 e^{4t}\begin{bmatrix} 0 \\ 1 \end{bmatrix} + c_2 e^{4t}\begin{bmatrix} 1 \\ 0 \end{bmatrix} = \begin{bmatrix} c_2\,e^{4t} \\ c_1\,e^{4t} \end{bmatrix}$.

For the Maple approach, we enter the equations:

```
> deq1:=diff(x(t),t)=4*x(t);
```

$$deq1 := \frac{d}{dt}\,x(t) = 4\,x(t)$$

```
> deq2:=diff(y(t),t)=4*y(t);
```

$$deq2 := \frac{d}{dt}\,y(t) = 4\,y(t)$$

We solve the system.

```
> soln:=dsolve({deq1,deq2},{x(t),y(t)});
```

$$soln := \left\{x(t) = \_C2\,e^{4t},\ y(t) = \_C1\,e^{4t}\right\}$$

We look at the direction field for the system in the x-y plane. Each point in this plane is of the form $\mathbf{r}(t) = (x(t), y(t))$. Each arrow represents the direction of the **velocity** vector $\mathbf{r}'(t) = (x'(t), y'(t))$ at $\mathbf{r}(t)$.

```
> DEplot ({deq1,deq2}, [x(t),y(t)], t=-10..10,x=-10..10,y=-10..10,
  arrows =medium,color=magenta,stepsize=0.05);
```



We now overlap a phase portrait on the direction field.

```
> p1:=plot(0,x=-10..10,y=-10..10,color=blue,thickness=3):
  p2:=pointplot([[0,-12],[0,12]],style=line,color=red,thickness=3):
  p3:=DEplot ({deq1,deq2}, [x(t),y(t)], t=-10..10,x=-10..10,y=-10.
  .10,arrows =medium,color=magenta,[seq([x(0)=cos(Pi/12*i),y(0)=sin
  (Pi/12*i)],i=1..24)],stepsize=0.05,linecolor=green):
  display(p1,p2,p3);
```



>

Each line and curve is a **trajectory** or **orbit** of a solution. Recall that x(t)=0, y(t)=0 is a solution of every homogeneous linear system. The path or trajectory of this solution is the single point at the origin. Because of the uniqueness of solutions, only one trajectory can go through each point in the plane. The red and blue lines represent the eigenspace E_4, which is two dimensional. The two blue solutions are parallel to the one of the eigenvectors for 4. Since the eigenvalue is positive, the solutions move away from the origin. The two red solutions are parallel to the other eigenvector for 4. Since the eigenvalue is positive, the solutions move away from the origin. The green trajectories are the trajectories of linear combinations of solutions from two linearly independent eigenvectors. We see that each green trajectory moves away from the origin. This gives us a good picture of the totality of solutions.

>

Repeated Eigenvalue with All Eigenvectors Linearly Dependent

```
> restart:with(LinearAlgebra):with(plots):with(DEtools):
```

We wish to solve the planar homogeneous linear system

$$\left[\begin{array}{l}\dfrac{dx}{dt} = x - y \\ \dfrac{dy}{dt} = x + 3\,y\end{array}\right] \text{ or } \mathbf{r}' = A\mathbf{r} \text{ with } A = \begin{bmatrix} 1 & -1 \\ 1 & 3 \end{bmatrix}.$$

We begin with the matrix A.

```
> A:=Matrix(2,2,[[1,-1],[1,3]]);
```

$$A := \begin{bmatrix} 1 & -1 \\ 1 & 3 \end{bmatrix}$$

We find its eigenvalues and corresponding eigenvectors.

```
> v:=Eigenvectors(A,output=list);
```

$$v := \left[\left[2, 2, \left\{\begin{bmatrix} -1 \\ 1 \end{bmatrix}\right\}\right]\right]$$

We see that 2 is the only eigenvalue and of multiplicity two with $\begin{bmatrix} -1 \\ 1 \end{bmatrix}$ as a corresponding eigenvector. Thus all eigenvectors are scalar multiples of this one. To get a second linearly independent vector, we need to find the vector **w** that is the solution of the vector equation (A - λI)**w** = **v**. We do this now.

```
> Id:=IdentityMatrix(2);
```

$$Id := \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$$

```
> w:=LinearSolve(<A-v[1,1]*Id|v[1,3,1]>);
```

$$w := \begin{bmatrix} 1 - \_t_2 \\ \_t_2 \end{bmatrix}$$

Thus we can choose **w** = $\begin{bmatrix} 1 \\ 0 \end{bmatrix}$. Then the general solution is $\mathbf{r} = c_1 e^{2t}\begin{bmatrix} -1 \\ 1 \end{bmatrix} + c_2\, e^{2t}\left(t\begin{bmatrix} -1 \\ 1 \end{bmatrix} + \begin{bmatrix} 1 \\ 0 \end{bmatrix}\right)$

$$= \begin{bmatrix} -c_1\, e^{2t} - c_2\,(1-t)\, e^{2t} \\ c_1\, e^{2t} + c_2\, t\, e^{2t} \end{bmatrix}.$$

For the Maple approach, we enter the equations:

```
> deq1:=diff(x(t),t)=x(t)-y(t);
```

$$deq1 := \frac{d}{dt}\, x(t) = x(t) - y(t)$$

```
> deq2:=diff(y(t),t)=x(t)+3*y(t);
```

$$deq2 := \frac{d}{dt}\, y(t) = x(t) + 3\, y(t)$$

We solve the system.

```
> soln:=dsolve({deq1,deq2},[x(t),y(t)]);
```

$$soln := \left\{x(t) = -e^{2t}\,(\_C1 + \_C2\, t - \_C2),\; y(t) = e^{2t}\,(\_C1 + \_C2\, t)\right\}$$

We look at the direction field for the system in the x-y plane. Each point in this plane is of the form $\mathbf{r}(t)$ = $(x(t), y(t))$. Each arrow represents the direction of the **velocity** vector $\mathbf{r}'(t)$ = $(x'(t), y'(t))$ at $\mathbf{r}(t)$.

```
> DEplot ({deq1,deq2}, [x(t),y(t)], t=-10..10,x=-10..10,y=-10..10,
  arrows =medium,color=magenta,stepsize=0.05);
```



We now overlap a phase portrait on the direction field.

```
> p1:=plot([[-exp(2*t),exp(2*t),t=-10..10],[exp(2*t),-exp(2*t),t=
  -10..10],[-t*exp(2*t),(t+1)*exp(2*t),t=-10..10],[t*exp(2*t),-(t+1)
  *exp(2*t),t=-10..10]],x=-10..10,y=-10..10,color=[red,red,blue,
  blue],thickness=3):
  p2:=DEplot ({deq1,deq2}, [x(t),y(t)], t=-10..10,x=-10..10,y=-10.
  .10,arrows =medium,color=magenta,[seq([x(0)=2*i,y(0)=0],i=1..5),
  seq([x(0)=-2*i,y(0)=0],i=1..5),seq([x(0)=0,y(0)=2*i],i=1..5),seq(
  [x(0)=0,y(0)=-2*i],i=1..5),seq([x(0)=0,y(0)=i/3],i=1..2),seq([x(0)
  =0,y(0)=-i/3],i=1..2)],stepsize=0.05,linecolor=green):
  display(p1,p2);
```



Each line and curve is a **trajectory** or **orbit** of a solution. Recall that x(t)=0, y(t)=0 is a solution of every homogeneous linear system. The path or trajectory of this solution is the single point at the origin. Because of the uniqueness of solutions, only one trajectory can go through each point in the plane. The red line represents two solutions from the eigenspace E_2, which is one dimensional, each moving away from the origin since 2 is positive. The two blue solutions are determined by an eigenvector of 2 along with the other linearly independent vector found. Since the eigenvalue 2 is positive, the solutions move away from the origin. The green trajectories again are the trajectories of linear combinations. We see that each green trajectory moves away from the origin. This gives us a good picture of the totality of solutions.

>